OMO: Software cost estimation

Tim Menzies

Lane Department of Computer Science, West Virginia University, PO Box 6109, Morgantown, WV, 26506-6109, USA;
http://tim.menzies.us; tim@menzies.us

Wp ref: ~menzies/src/pl/prod/omo.pl, March 7, 2003.

Abstract COCOMO is a software effort estimation tool. OMO is COCOMO written in SWI-Prolog [16]

Contents

List of Figures

1 What is COCOMO?

The COCOMO project aims at developing an open-source, public-domain software effort estimation model. The project has collected information on 161 projects from commercial, aerospace, government, and non-profit organizations [1, 15]. As of 1998, the projects represented in the database were of size 20 to 2000 KSLOC (thousands of lines of code) and took between 100 to 10000 person months to build.

COCOMO measures effort in calendar months where one month is 152 hours (and includes development and management hours). The core intuition behind COCOMO-based estimation is that as systems grow in size, the effort required to create them grows exponentially, i.e. $effort \propto KSLOC^x$. More precisely:

$$months = a * \left(KSLOC^{\left(0.91+\sum_{i=1}^{5} SF_i\right)}\right) * \left(\prod_{j=1}^{17} EM_j\right)$$

where a is a domain-specific parameter, and KSLOC is estimated directly or computed from a function point analysis. SF_i are the scale factors (e.g. factors such as “have we built this kind of system before?”) and EM_j are the cost drivers (e.g. required level of reliability). Figure 1 lists the scale drivers and effort multipliers.

Software effort-estimation models like COCOMO-II should be tuned to their local domain. Off-the-shelf “untuned” models have been up to 600% inaccurate in their estimates, e.g. [14, p165] and [2]. However, tuned models can be far more accurate. For example, [1] reports a study with a bayesian tuning algorithm using the COCOMO project database. After bayesian tuning, a cross-validation study showed that COCOMO-II model produced estimates that are within 30% of the actuals, 69% of the time.

Figure 2 shows the sizes of various COCOMO tuning parameters. Notice the linear fits of the top two tunings: these were generated via linear regression and hence are straight lines. The bottom row shows tunings generated from a genetic algorithm (GA): such GAs were designed to handle non-linear situations so their curve fits can be all over the place.

The intuition to be gained from Figure 2 is that some COCOMO parameters are more influential than others. Some

Type	Acronym	Definition	Low-end	Medium	High-end
EM	acap	analyst capability	worst 15%	55%	best 10%
EM	aexp	applications experience	2 months	1 year	6 years
SF	arch	architecture or risk resolution	few interfaces defined or few risk eliminated	most interfaces defined or most risks eliminated	all interfaces defined or all risks eliminated
EM	cplx	product complexity	e.g. simple read/write statements	e.g. use of simple interface widgets	e.g. performance-critical embedded systems
EM	data	database size (DB bytes/ Program SLOC)	10	100	1000
EM	docu	documentation	many life-cycle phases not documented		extensive reporting for each life-cycle phase
SF	flex	development flexibility	development process rigorously defined	some guidelines, which can be relaxed	only general goals defined
EM	ltex	language and tool-set experience	2 months	1 year	6 years
EM	pcap	programmer capability	worst 15%	55%	best 10%
EM	pcon	personnel continuity (% turnover per year)	48%	12%	3%
EM	pexp	platform experience	2 months	1 year	6 years
SF	pmat	process maturity	CMM level 1	CMM level 3	CMM level 5
SF	prec	precedentedness	we have never built this kind of software before	somewhat new	thoroughly familiar
EM	pvol	platform volatility ($\frac{frequency\ of\ major\ changes}{frequency\ of\ minor\ changes}$)	$\frac{12\ months}{1\ month}$	$\frac{6\ months}{2\ weeks}$	$\frac{2\ weeks}{2\ days}$
EM	rely	required reliability	errors mean slight inconvenience	errors are easily recoverable	errors can risk human life
EM	ruse	required reuse	none	across program	across multiple product lines
EM	sced	dictacted development schedule	deadlines moved closer to 75% of the original estimate	no change	deadlines moved back to 160% of the original estimate
EM	site	multi-site development	some contact: phone, mail	some email	interactive multi-media
EM	stor	main storage constraints (% of available RAM)	N/A	50%	95%
SF	team	team cohesion	very difficult interactions	basically co-operative	seamless interactions
EM	time	execution time constraints (% of available CPU)	N/A	50%	95%
EM	tool	use of software tools	edit,code,debug		well intergrated with lifecycle

Fig. 1 Parameters of the COCOMO-II effort risk model; adapted from `http://sunset.usc.edu/COCOMOII/expert_cocomo/drivers.html`. "`Stor`" and "`time`" score "N/A"" for low-end values since they have no low-end defined in COCOMO-II. "SF" denotes "scale factors" and "EM" denotes "effort multipliers".

are weakly correlated to increasing effort (column 1); some are weakly correlated to decreasing effort (column 2); and some are strongly correlated to decreasing effort (column 3). This will be useful later when we write search engines to control COCOMO. A core heuristic will be "change the influential parameters first".

The last column of Figure 2 relate to the effort multipliers. While shown here as linear, their influence can be even greater than that since they are used up in an exponential equation.

2 Requires

```
:- load_files([lib        % grab standard stuff [10]
              ,cfg        % options controller [11]
              ,gpl0, gpl1 % GPL-2 license stuff [7]
              ,omo0       % pre-load actions
              ,omolib     % local libraries
              ,omo1       % predicates
              ,omo2       % start-up commands
              ,ufp2sloc   % function points per LOC database
              ],[silent(yes),if(changed)]).
```



Fig. 2 Influence of different COCOMO parameters

3 Pre-load actions

3.1 Operators

Something to mark decisions we are considering

```
:- op(1,xf,?).
```

Something to mark range declarations.

```
:- op(701,xfx,of).
```

3.2 Flags

```
:- discontiguous (of/3),(range/3),(goal/2).
:- dynamic (of/3),(range/3),(goal/2).
```

3.3 Hooks

Define some expected value, and its associated range predicate.

```
term_expansion(X of Y,Z) :- ofs(X of Y,Z).
```

4 Main System

4.1 Main driver

```

estimate :- project := P, estimate(P).

estimate(P) :-
    getProject(P),
    cocomo(Coc),
    estimate(Pm,Staff,Months),
    format('COCOMO.~p says ~p months (total);',[Coc,Pm]),
    format('~p staff over ~p months\n', [Staff,Months]).

estimate(Pm,Staff,Months) :-
    tdev(Tdev),
    pm(Pm0),
    Pm is Pm0,
    Staff is ceiling(Pm/Tdev),
    Months is ceiling(Tdev),
    !.
```

4.2 Equations

4.2.1 Sizing equations

```
size((1 + (R/100)) *(N + E)) :-
    revl(R), newKsloc(N), equivalentKsloc(E).

equivalentKsloc(Ak*Aam*(1-(At/100))) :-
    adaptedKsloc(Ak), at(At), aam(Aam).

aam(Am) :- aaf(Af), compare(C,Af,50), aam1(C,Af,Am).

aam1(=,Af, X) :- aam1(<,Af,X).
aam1(>,Af, (Aa+Af*(Su*U))/100) :- aa(Aa),su(Su),unfm(U).
aam1(<,Af,((Aa+Af*(1+(0.02*Su*U)))/100)) :-
    aa(Aa), su(Su), unfm(U).

aaf(0.4*Dm+0.3*Cm+0.3*Im) :- dm(Dm), cm(Cm), im(Im).
```

4.2.2 Schedule Equations

```
tdev((C*(P^F))*SP/100) :-
    c(C), pmNs(P), f(F), scedPercent(SP).

f( D + 0.2*(E-B))  :-
    d(D),e(E), b(B).
```

4.2.3 Effort Equations hmmm... sced value never used

```
pm(Pm0*Em17+Pa) :-
    pmNs(Pm0), w(sced,Em17), pmAuto(Pa).

pmNs(A*(S^E)*Em1 *Em2 *Em3 *Em4 *Em5 *Em6 *Em7*Em8*Em9*
            Em10*Em11*Em12*Em13*Em14*Em15*Em16) :-
    a(A), size(S), e(E), w(rely,Em1), w(data,Em2),
    w(cplx,Em3), w(ruse,Em4), w(docu,Em5), w(time,Em6),
    w(stor,Em7), w(pvol,Em8), w(acap,Em9), w(pcap,Em10),
    w(pcon,Em11),w(aexp,Em12),w(pexp,Em13),
    w(ltex,Em14),w(tool,Em15),w(site,Em16).

e(B + 0.01*(Sf1+Sf2+Sf3+Sf4+Sf5)) :-
    b(B),
    w(prec,Sf1), w(flex,Sf2),w(arch,Sf3),
    w(team,Sf4), w(pmat,Sf5).

pmAuto((Ak*(At/100))/Ap) :-
    adaptedKsloc(Ak), at(At), atKprod(Ap).
```

4.3 Tunings

4.3.1 Constants

```
a(2.5)  :- cocomo(1983).
a(2.94) :- cocomo(2000).
a(2.94) :- cocomo(ga).

b(0.91) :- cocomo(2000).
b(1.01) :- cocomo(1983).
b(1.01) :- cocomo(ga).

c(3.0)  :- cocomo(1983).
c(3.67) :- cocomo(2000).
c(3.67) :- cocomo(ga).

d(0.28) :- cocomo(2000).
d(0.33) :- cocomo(1983).
d(0.33) :- cocomo(ga).
```

4.3.2 Post-architecture scale factors The COCOMO 2000 scale factors learnt via bayesian tuning.

```
postArch(2000,scaleFactors) has
           [ xl, vl,   l,   n,   h,  vh,  xh]+
    [[prec,  _,6.20,4.96,3.72,2.48,1.24,   _]
    ,[flex,  _,5.07,4.05,3.04,2.03,1.01,   _]
    ,[arch,  _,7.07,5.65,4.24,2.83,1.41,   _]
    ,[team,  _,5.48,4.38,3.29,2.19,1.01,   _]
    ,[pmat,  _,7.80,6.24,4.68,3.12,1.56,   _]
    ].
```

The original scale factors.

```
postArch(1983,scaleFactors) has
           [ xl,  vl,   l,   n,   h,  vh, xh]+
    [[prec,   _,4.05,3.24,2.43,1.62,0.81,  _]
    ,[flex,   _,6.07,4.86,3.64,2.43,1.21,  _]
    ,[arch,   _,4.22,3.38,2.53,1.69,0.84,  _]
    ,[team,   _,4.94,3.95,2.97,1.98,0.99,  _]
    ,[pmat,   _,4.54,3.64,2.73,1.82,0.91,  _]
    ].

aa :- expand_term(postArch(1983,scaleFactors) has
           [ xl,  vl,   l,   n,   h,  vh, xh]+
    [[prec,   _,4.05,3.24,2.43,1.62,0.81,  _]
    ,[flex,   _,6.07,4.86,3.64,2.43,1.21,  _]],
    L),
    writes(L).
```

Some scale factors learnt via some genetic algorithms.

```
postArch(ga,scaleFactors) has
            [ xl,  vl,   l,   n,   h,  vh,  xh]+
    [[prec,   _,4.05,3.24,2.43,1.62,0.81,    _]
    ,[flex,    _,6.07,4.86,3.64,2.43,1.21,   _]
    ,[arch,    _,4.22,3.38,2.53,1.69,0.84,   _]
    ,[team,    _,4.94,3.95,2.97,1.98,0.99,   _]
    ,[pmat,    _,4.54,3.64,2.73,1.82,0.91,   _]
    ].
```

4.3.3 Post-architecture effort multipliers: The COCOMO 2000 effort multipliers learnt via bayesian tuning.

```
postArch(2000,effortMultiplers) has
          [xl,   vl,   l,   n,  h,  vh,  xh]+
    [[rely,  _,0.82,0.92,1.00,1.10,1.26,   _]
    ,[data,  _,_   ,0.90,1.00,1.14,1.28,   _]
    ,[cplx,  _,0.73,0.87,1.00,1.17,1.34,1.74]
    ,[ruse,  _,_   ,0.95,1.00,1.07,1.15,1.24]
    ,[docu,  _,0.81,0.91,1.00,1.11,1.23,   _]
    ,[time,  _,   _,   _,1.00,1.11,1.29,1.63]
    ,[stor,  _,   _,   _,1.00,1.05,1.17,1.46]
    ,[pvol,  _,   _,0.87,1.00,1.15,1.30,   _]
    ,[acap,  _,1.42,1.19,1.00,0.85,0.71 ,  _]
    ,[pcap,  _,1.34,1.15,1.00,0.88,0.76,   _]
    ,[pcon,  _,1.29,1.12,1.00,0.90,0.81,   _]
    ,[aexp,  _,1.22,1.10,1.00,0.88,0.81,   _]
    ,[pexp,  _,1.19,1.09,1.00,0.91,0.85,   _]
    ,[ltex,  _,1.20,1.09,1.00,0.91,0.84,   _]
    ,[tool,  _,1.17,1.09,1.00,0.90,0.78,   _]
    ,[site,  _,1.22,1.09,1.00,0.93,0.86,0.80]
    ,[sced,  _,1.43,1.14,1.00,1.00,1.00,   _]
    ].
```

The original effort multipliers.

```
postArch(1983,effortMultiplers) has
           [ xl,  vl,   l,   n,   h,  vh,  xh]+
    [[rely,   _,0.75,0.88,1.00,1.15,1.40,   _]
    ,[data,   _,   _,0.94,1.00,1.08,1.16,   _]
    ,[cplx,   _,0.75,0.88,1.00,1.15,1.30,1.65]
    ,[ruse,   _,   _,0.89,1.00,1.16,1.34,1.56]
    ,[docu,   _,0.85,0.93,1.00,1.08,1.17,   _]
    ,[time,   _,   _,   _,1.00,1.11,1.30,1.66]
    ,[stor,   _,   _,   _,1.00,1.06,1.21,1.56]
    ,[pvol,   _,   _,0.87,1.00,1.15,1.30,   _]
    ,[acap,   _,1.50,1.22,1.00,0.83,0.67,   _]
    ,[pcap,   _,1.37,1.16,1.00,0.87,0.74,   _]
    ,[pcon,   _,1.26,1.11,1.00,0.91,0.83,   _]
    ,[aexp,   _,1.23,1.10,1.00,0.88,0.80,   _]
    ,[pexp,   _,1.26,1.12,1.00,0.88,0.80,   _]
    ,[ltex,   _,1.24,1.11,1.00,0.90,0.82,   _]
    ,[tool,   _,1.20,1.10,1.00,0.88,0.75,   _]
    ,[site,   _,1.24,1.10,1.00,0.92,0.85,0.79]
    ,[sced,   _,1.23,1.08,1.00,1.04,1.10,   _]
    ].
```

Some effort multipliers learnt via some genetic algorithms.

```
postArch(ga,effortMultiplers) has
           [ xl,  vl,   l,   n,   h,  vh,  xh]+
    [[rely,    _,0.79,0.78,1.00,1.16,1.41,   _]
    ,[data,    _,   _,0.96,1.00,1.31,1.20,   _]
    ,[cplx,    _,0.90,1.06,1.00,0.99,0.99,0.87]
    ,[ruse,    _,   _,0.89,1.00,1.16,1.34,1.56]
    ,[docu,    _,0.85,0.93,1.00,1.08,1.17,   _]
    ,[time,    _,   _,   _,1.00,1.01,1.24,2.13]
    ,[stor,    _,   _,   _,1.00,1.36,1.37,1.42]
    ,[pvol,    _,   _,1.25,1.00,1.13,1.15,   _]
    ,[acap,    _,1.19,1.26,1.00,1.00,0.73,   _]
    ,[pcap,    _,1.71,1.73,1.00,0.75,0.74,   _]
    ,[pcon,    _,1.26,1.11,1.00,0.91,0.83,   _]
    ,[aexp,    _,1.41,1.02,1.00,0.64,0.86,   _]
    ,[pexp,    _,1.26,1.12,1.00,0.88,0.80,   _]
    ,[ltex,    _,1.24,1.11,1.00,0.90,0.82,   _]
    ,[tool,    _,1.13,0.91,1.00,1.09,2.86,   _]
    ,[site,    _,1.24,1.10,1.00,0.92,0.85,0.79]
    ,[sced,    _,1.22,1.29,1.00,0.72,0.29,   _]
    ].
```

4.4 Data dictionary

4.4.1 General

```
languageP(X) :- upf2sloc(X,_).

sym(X) :- rsym(X).

onezeroP(X) :- rin(0,1,0.2,X), number(X).

percentP(X) :- rin(0,100,1,X),integer(X).

posint(X)   :- rin(0,65536,X),integer(X).
posnum(X)   :- rin(0,inf,X),number(X).

num10(X) :- rin(0,10,X), number(X).

cocomoP(2000).
cocomoP(1983).
cocomoP(ga).

vlvh(n). vlvh(l). vlvh(h). vlvh(vl). vlvh(vh).

lvh(n). lvh(l). lvh(h). lvh(vh).

vlxh(n).  vlxh(l).  vlxh(h).
vlxh(vl). vlxh(vh). vlxh(xh).

lxh(n). lxh(l). lxh(h). lxh(vh). lxh(xh).

nxh(n). nxh(h). nxh(vh). nxh(xh).
```

4.4.2 Specifics Declare what ranges are appropriate for what variables.

```
cocomo       of cocomoP.    label        of sym.
language     of languageP.  revl         of percentP.
newKsloc     of percentP.   adaptedKsloc of posint.
cm           of percentP.   dm           of percentP.
im           of percentP.   aa           of percentP.
unfm         of onezeroP.   su           of percentP.
at           of percentP.   atKprod      of posnum.
scedPercent  of percentP.

prec of vlvh.  flex of vlvh. arch of vlvh.
team of vlvh.  pmat of vlvh. rely of vlvh.
data of lvh.   cplx of vlxh. ruse of lxh.
docu of vlvh.  time of nxh.  stor of nxh.
pvol of lvh.   acap of vlvh. pcap of vlvh.
pcon of vlvh.  aexp of vlvh. pexp of vlvh.
ltex of vlvh.  tool of vlvh. site of vlxh.
sced of vlvh.
```

5 OMO Support code

5.1 `GetProject/1` *zaps old project knowledge*

Definitions of assertions created when projects are loaded.

```
defProj(range(_,_,_)).
defProj(option(X,_)) :- of(X,_,_).
defProj(goal(_,_)).

getProject(X) :-
    proj0,          % project details now dynamic
    projReset,      % zap old details
    [X],            % load projects details
    readies(Items), % find side-effects
    forall(member(One,Items),
           getProject1(One)).

getProject1((:- X)) :- !,X.
getProject1(X) :- assert(X).
```

Support code for the above:

```
proj0 :- all
         defProj(T),
     functor(T,F,A),
     dynamic(F/A),
     discontiguous(F/A).

projReset :-
    all defProj(T), (retract(T); retract((T :- _))).
```

5.2 Defining expected variables.

The assertion "`Var of Pred.`" gives OMO the expectation that the predicate `Pred(Value)` can be used to check supplied values for `Var`. Alternatively, if none are generated, then `Pred(Value)` can be used to generate a value for `Var`.

Internally, "`Var of Pred.`" is stored in an `of/3` assertion:

```
of(Var,Pred(Value),Value).
```

```
ofs(A,Bs) :-
    bagof(B,A^of1(A,B),Bs).

of1(X of Y,_) :-
    \+ ground(X of Y),
    !,barphln(mustBeLowerCase(X of Y)).
of1(_ of Y,_) :-
    Pred=.. [Y,_],
    \+ clause(Pred,_),
    !,barphln(unknownType(Y)).
of1(X of Y,Out) :-
    Head=.. [X,Value],
    Body=.. [Y,Value],
    (Out=(Head :- range(X,_,Value),Body)
        ;Out=of(X,Body,Value)).
```

General form	Notes	Example
`Var = List?`	`Var` can take on the variables in `List`. mark all items in `List` as goals	`cplx = [vh,xh]?`.
`Var=[X1,X2?,X3,..]`	`Var` can take any of the variables in the list. Some of these values are goal values.	`ruse =[l,n,h?].`
`Var= [X1,X2,X3,..]`	`Var` can take any of the variables in the list. None of these values are goal values.	`time = [n,h,vh].`
`Var= X?`	`Var` can take only take one value. and that value is a goal.	`pvol = h?.`
`Var= ?`	`Var` can take any value over its range and all of those values are goals.	`data = ? .`
`Var= X`	`Var` can take one value and that value is not agoal.	`pcap = n.`

Fig. 3 Specifying ranges for variables.

5.3 Checking supplied variables

(Assumes `of/3` facts has been previously generated).

Users of this system can supply values to be used in the simuation. That input description includes mostly *range* values and a few *goal* values. Simulations backtrack over the *range* of values. Optimizers of this simulation can query the *goal* values to constrain their optimizations to just the *goal*s.

Syntactically, goal values are marked with a question mark (e.g. `X?`) and anything not marked in this way is a range value. Figure 3 shows the various forms.

Via a `term_expansion`, the `ready` assertion triggers the following code;

```
readies(L) :- bagof(X,ready(X),L).
```

For variables with no settings, use the `of/3`'s `Pred` to get a value.

```
ready((range(X,n,Value) :- Pred)) :-
    of(X,Pred,Value),
    \+ option(X,_).
```

Complain if a variable's setting is illegal.

```
ready((:- burp(badValue(W,Value)))) :-
    of(W,Pred,Value),
    option(W,X),
    once(ready0(X,_,_,Y)),
    member(mark(_,Value),Y), %range,goal,guess
    \+ Pred.
```

Otherwise, generate appropiate `range`

```
ready(Out):-
    of(W,Pred,Value),
    option(W,X),
    once(ready0(X,Pred,Value,Y)),
    member(Z,Y),
    ready1(Z,W,Out).
```

```
ready0([H|T]?, _,_,L) :- maplist(ready0aGoal,[H|T],L).
ready0([H|T],  _,_,L) :- maplist(ready0a,[H|T],L).
ready0(Item?,  _,_,[mark(goal,Item)]).
ready0( ?   ,  P,V,[pred(P,V)]).
ready0(Item,   _,_,[mark(range,Item)]).

ready0a(X?,mark(goal,X)).
ready0a(X, mark(guess,X)) :- atomic(X).

ready0aGoal(X,mark(guess,X)).
```

Anything marked as a range generates a **range** fact.

```
ready1(mark(range,Y),   X, range(X,1,Y)).
```

Anything that is a guess is one of the **range**s we want to guess.

```
ready1(mark(guess,Y),   X, range(X,n,Y)).
```

Anything marked as a goal generated a **range** and a **goal** fact.

```
ready1(mark(goal,One), X, range(X,n,One)).
ready1(mark(goal,One), X, goal(X,One)).
```

Anything marked as a goal of unknwon range generates range and goal rules which pull all values from the predicate **P**.

```
ready1(pred(P,V),      X, (range(X,n,V) :- P)).
ready1(pred(P,V),      X, (goal(X,V)  :- P)).
```

5.4 w/2

Convert scores to numeric weights

```
w(A,W) :-
    range(A,_,S),
    postArch(A,S,W),
    num10(W).

postArch(A,S,W) :-
    cocomo(When),
    call(lookUp(postArch(When,_),A,S,W)).
```

6 Knowledge base

6.1 Sample project "eg1"

```
cocomo       = ga.
label        =  'eg1'.
language     =  prolog.
revl         =  10.
newKsloc     = 100.
adaptedKsloc =   0.
cm           =   0.   % new code
dm           =   0.   % new code
im           =   0.   % new code
aa           =   2.   % basic module search + docu [15, p24]
unfm         =   0.4. % somewhat familiar
su           =  30.   % nom=al value [15, p23]
at           =   0.
atKprod      =   2.4.
scedPercent  = 100.
prec         = [vl,l].
flex         = [l?,n,h,vh].
arch         = [vl,l, n?].
team         = [l, n?].
pmat         = [vl,l,n, h?].
rely         = vh.
data         = n .
cplx         = [vh,xh]? .
ruse         = [l,n,h?].
docu         = [l,n,h?].
time         = [n,h,vh].
stor         = [n?,h,vh].
pvol         = l.
acap         = [l,n?].
pcap         = n.
pcon         = [l,n?].
aexp         = [l,n].
pexp         = [n].
ltex         = [l,n,h?].
tool         = [l,n].
site         = n.
sced         = [vl,l,n?].
```

Important!: all project descriptions have to end with the "**ready.**" assertion.

```
ready.
```

7 Start-up actions

Usual stuff.

```
:- sneak(
        ['defaults.omo' % see Figure ??
        ,'config.omo'   % see Figure ??
        ,ufp2sloc       % see §??
        ]).

:- commandLine.
:- ?verbose -> hello ; true.
```

8 Bugs

None known but many suspected.

Acknowledgements This research was conducted at West Virginia University under NASA contract NCC2-0979 and NCC5-685. The work was sponsored by the NASA Office of Safety and Mission Assurance under the Software Assurance Research Program led by the NASA IV&V Facility. Reference herein to any specific commercial product, process, or service by trade name, trademark, manufacturer, or otherwise, does not constitute or imply its endorsement by the United States Government.

References

1. S. Chulani, B. Boehm, and B. Steece. Bayesian analysis of empirical software engineering cost models. *IEEE Transaction on Software Engineerining*, 25(4), July/August 1999.
2. C. Kemerer. An empirical validation of software cost estimation models. *Communications of the ACM*, 30(5):416–429, May 1987.
3. T. Menzies. PROD: A PROlog Documentation, and Delivery tool, 2003. Available from `http://tim.menzies.com/pdf/03prod2.pdf`.
4. T. Menzies. PROD: the soruce code, 2003. Available from `http://tim.menzies.com/prodfiles.zip`.
5. T. Menzies. An example of the PROD prolog delivery and documentation system, 2003. Available from `http://tim.menzies.com/pdf/03prod1.pdf`.
6. T. Menzies. A family database: documentation of a very simple prolog family database using PROD., 2003. Available from `http://tim.menzies.com/pdf/03family.pdf`.
7. T. Menzies. Including gpl-2 in PROD, 2003. Available from `http://tim.menzies.com/pdf/03gpl.pdf`.
8. T. Menzies. Monte carlo simulations in prolog: a PROD tool, 2003. Available from `http://tim.menzies.com/pdf/03lurch.pdf`.
9. T. Menzies. Motivations: the why and who of PROD, 2003. Available from `http://tim.menzies.com/pdf/03prodabout.pdf`.
10. T. Menzies. PROD's commonly used predicates, 2003. Available from `http://tim.menzies.com/pdf/03lib.pdf`.
11. T. Menzies. PROD's handler for config files and command line options, 2003. Available from `http://tim.menzies.com/pdf/03cfg.pdf`.
12. T. Menzies. Software cost estimation: a PROD tool, 2003. Available from `http://tim.menzies.com/pdf/03omo.pdf`.
13. T. Menzies. Title: a bare-bones minimal example of PROD., 2003. Available from `http://tim.menzies.com/pdf/03prod3.pdf`.
14. T. Mukhopadhyay, S. Vicinanza, and M. Prietula. Examining the feasibility of a case-based reasoning tool for software effort estimation. *MIS Quarterly*, pages 155–171, June 1992.
15. B. W.Boehm, E. Horowitz, R. Madachy, D. Reifer, B. K. Clark, B. Steece, A. W. Brown, S. Chulani, and C. Abts. *Software Cost Estimation with Cocomo II*. Prentice Hall, 2000.
16. J. Wielemaker. *SWI-Prolog*. Available from `http://swi.psy.uva.nl/projects/xpce/SWI-Prolog.html`.

A License

This software is distributed under the GNU General Public License.

A.1 nowarranty.txt

OMO comes with ABSOLUTELY NO WARRANTY: for more details type 'warranty'.

This is free software, and you are welcome to redistribute it under certain conditions: for more details, type 'conditions'.

This paper was prepared using the PROD Prolog documenet and delivery preperation system [3, 4]:

- PROD was created to help me train my graduate students. It also is a nice demonstration of the utility of Prolog. For more on this point, see [9] (warning: includes sermonizing).
- PROD comes with a simple examples of documenting Prolog source code (e.g. a predicate to sum a list of numbers [5] and the standard Prolog family database [6] as well as a "bare-bones" example [13].
- PROD was designed as an open source tool. The GNU public license (version 2) can be shown from any PROD application and is added to every PROD document [7]).
- PROD comes with some handy little utilities and some applications:
  - A handler for config files, user preferences, and command line options [11].
  - A library of commonly used predicates [10].
  - A software cost estimation tool [12].
  - A monte carlo simulation package for Prolog [8].

Fig. 4 This document is part of the PROD delivery and documentation tool for Prolog applications. To find out more about PROD, the best place to start is [3].

A.2 warranty.txt

This program is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; version 2 (see http://www.gnu.org/copyleft/gpl.html or type 'conditions').

This program is distributed in the hope that it will be useful, but WITHOUT ANY WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU General Public License for more details.

You should have received a copy of the GNU General Public License along with this program; if not, write to the Free Software Foundation, Inc., 59 Temple Place - Suite 330, Boston, MA 02111-1307, US.

A.3 conditions.txt

TERMS AND CONDITIONS FOR COPYING, DISTRIBUTION AND MODIFICATION

0. This License applies to any program or other work which contains a notice placed by the copyright holder saying it may be distributed under the terms of this General Public License. The 'Program', below, refers to any such program or work, and a 'work based on the Program' means either the Program or any derivative work under copyright law: that is to say, a work containing the Program or a portion of it, either verbatim or with modifications and/or translated into another language. (Hereinafter, translation is included without limitation in the term 'modification'.) Each licensee is addressed as 'you'.

Activities other than copying, distribution and modification are not covered by this License; they are outside its scope. The act of running the Program is not restricted, and the output from the Program is covered only if its contents constitute a work based on the Program (independent of having been made by running the Program). Whether that is true depends on what the Program does.

1. You may copy and distribute verbatim copies of the Program's source code as you receive it, in any medium, provided that you conspicuously and appropriately publish on each copy an appropriate copyright notice and disclaimer of warranty; keep intact all the notices that refer to this License and to the absence of any warranty; and give any other recipients of the Program a copy of this License along with the Program.

You may charge a fee for the physical act of transferring a copy, and you may at your option offer warranty protection in exchange for a fee.

2. You may modify your copy or copies of the Program or any portion of it, thus forming a work based on the Program, and copy and distribute such modifications or work under the terms of Section 1 above, provided that you also meet all of these conditions:

a) You must cause the modified files to carry prominent notices stating that you changed the files and the date of any change.

b) You must cause any work that you distribute or publish, that in whole or in part contains or is derived from the Program or any part thereof, to be licensed as a whole at no charge to all third parties under the terms of this License.

c) If the modified program normally reads commands interactively when run, you must cause it, when started running for such interactive use in the most ordinary way, to print or display an announcement including an appropriate copyright notice and a notice that there is no warranty (or else, saying that you provide a warranty) and that users may redistribute the program under these conditions, and telling the user how to view a copy of this License. (Exception: if the Program itself is interactive but does not normally print such an announcement, your work based on the Program is not required to print an announcement.)

These requirements apply to the modified work as a whole. If identifiable sections of that work are not derived from the Program, and can be reasonably considered independent and separate works in themselves, then this License, and its terms, do not apply to those sections when you distribute them as separate works. But when you distribute the same sections as part of a whole which is a work based on the Program, the distribution of the whole must be on the terms of this License, whose permissions for other licensees extend to the entire whole, and thus to each and every part regardless of who wrote it.

Thus, it is not the intent of this section to claim rights or contest your rights to work written entirely by you; rather, the intent is to exercise the right to control the distribution of derivative or collective works based on the Program.

In addition, mere aggregation of another work not based on the Program with the Program (or with a work based on the Program) on a volume of a storage or distribution medium does not bring the other work under the scope of this License.

3. You may copy and distribute the Program (or a work based on it, under Section 2) in object code or executable form under the terms of Sections 1 and 2 above provided that you also do one of the following:

a) Accompany it with the complete corresponding machine-readable source code, which must be distributed under the terms of Sections 1 and 2 above on a medium customarily used for software interchange; or,

b) Accompany it with a written offer, valid for at least three years, to give any third party, for a charge no more than your cost of physically performing source distribution, a complete machine-readable copy of the corresponding source code, to be distributed under the terms of Sections 1 and 2 above on a medium customarily used for software interchange; or,

c) Accompany it with the information you received as to the offer to distribute corresponding source code. (This alternative is allowed only for noncommercial distribution and only if you received the program in object code or executable form with such an offer, in accord with Subsection b above.)

The source code for a work means the preferred form of the work for making modifications to it. For an executable work, complete source code means all the source code for all modules it contains, plus any associated interface definition files, plus the scripts used to control compilation and installation of the executable. However, as a special exception, the source code distributed need not include anything that is normally distributed (in either source or binary form) with the major components (compiler, kernel, and so on) of the operating system on which the executable runs, unless that component itself accompanies the executable.

If distribution of executable or object code is made by offering access to copy from a designated place, then offering equivalent access to copy the source code from the same place counts as distribution of the source code, even though third parties are not compelled to copy the source along with the object code.

4. You may not copy, modify, sublicense, or distribute the Program except as expressly provided under this License. Any attempt otherwise to copy, modify, sublicense or distribute the Program is void, and will automatically terminate your rights under this License. However, parties who have received copies, or rights, from you under this License will not have their licenses terminated so long as such parties remain in full compliance.

5. You are not required to accept this License, since you have not signed it. However, nothing else grants you permission to modify or distribute the Program or its derivative works. These actions are prohibited by law if you do not accept this License. Therefore, by modifying or distributing the Program (or any work based on the Program), you indicate your acceptance of this License to do so, and all its terms and conditions for copying, distributing or modifying the Program or works based on it.

6. Each time you redistribute the Program (or any work based on the Program), the recipient automatically receives a license from the original licensor to copy, distribute or modify the Program subject to these terms and conditions. You may not impose any further restrictions on the recipients' exercise of the rights granted herein. You are not responsible for enforcing compliance by third parties to this License.

7. If, as a consequence of a court judgment or allegation of patent infringement or for any other reason (not limited to patent issues), conditions are imposed on you (whether by court order, agreement or otherwise) that contradict the conditions of this License, they do not excuse you from the conditions of this License. If you cannot distribute so as to satisfy simultaneously your obligations under this License and any other pertinent obligations, then as a consequence you may not distribute the Program at all. For example, if a patent license would not permit royalty-free redistribution of the Program by all those who receive copies directly or indirectly through you, then the only way you could satisfy both it and this License would be to refrain entirely from distribution of the Program.

If any portion of this section is held invalid or unenforceable under any particular circumstance, the balance of the section is intended to apply and the section as a whole is intended to apply in other circumstances.

It is not the purpose of this section to induce you to infringe any patents or other property right claims or to contest validity of any such claims; this section has the sole purpose of protecting the integrity of the free software distribution system, which is implemented by public license practices. Many people have made generous contributions to the wide range of software distributed through that system in reliance on consistent application of that system; it is up to the author/donor to decide if he or she is willing to distribute software through any other system and a licensee cannot impose that choice.

This section is intended to make thoroughly clear what is believed to be a consequence of the rest of this License.

8. If the distribution and/or use of the Program is restricted in certain countries either by patents or by copyrighted interfaces, the original copyright holder who places the Program under this License may add an explicit geographical distribution limitation excluding those countries, so that distribution is permitted only in or among countries not thus excluded. In such case, this License incorporates the limitation as if written in the body of this License.

9. The Free Software Foundation may publish revised and/or new versions of the General Public License from time to time. Such new versions will be similar in spirit to the present version, but may differ in detail to address new problems or concerns.

Each version is given a distinguishing version number. If the Program specifies a version number of this License which applies to it and 'any later version', you have the option of following the terms and conditions either of that version or of any later version published by the Free Software Foundation. If the Program does not specify a version number of this License, you may choose any version ever published by the Free Software Foundation.

10. If you wish to incorporate parts of the Program into other free programs whose distribution conditions are different, write to the author to ask for permission. For software which is copyrighted by the Free Software Foundation, write to the Free Software Foundation; we sometimes make exceptions for this. Our decision will be guided by the two goals of preserving the free status of all derivatives of our free software and of promoting the sharing and reuse of software generally.

NO WARRANTY

11. BECAUSE THE PROGRAM IS LICENSED FREE OF CHARGE, THERE IS NO WARRANTY FOR THE PROGRAM, TO THE EXTENT PERMITTED BY APPLICABLE LAW. EXCEPT WHEN OTHERWISE STATED IN WRITING THE COPYRIGHT HOLDERS AND/OR OTHER PARTIES PROVIDE THE PROGRAM 'AS IS' WITHOUT WARRANTY OF ANY KIND, EITHER EXPRESSED OR IMPLIED, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE. THE ENTIRE RISK AS TO THE QUALITY AND PERFORMANCE OF THE PROGRAM IS WITH YOU. SHOULD THE PROGRAM PROVE DEFECTIVE, YOU ASSUME THE COST OF ALL NECESSARY SERVICING, REPAIR OR CORRECTION.

12. IN NO EVENT UNLESS REQUIRED BY APPLICABLE LAW OR AGREED TO IN WRITING WILL ANY COPYRIGHT HOLDER, OR ANY OTHER PARTY WHO MAY MODIFY AND/OR REDISTRIBUTE THE PROGRAM AS PERMITTED ABOVE, BE LIABLE TO YOU FOR DAMAGES, INCLUDING ANY GENERAL, SPECIAL, INCIDENTAL OR CONSEQUENTIAL DAMAGES ARISING OUT OF THE USE OR INABILITY TO USE THE PROGRAM (INCLUDING BUT NOT LIMITED TO LOSS OF DATA OR DATA BEING RENDERED INACCURATE OR LOSSES SUSTAINED BY YOU OR THIRD PARTIES OR A FAILURE OF THE PROGRAM TO OPERATE WITH ANY OTHER PROGRAMS), EVEN IF SUCH HOLDER OR OTHER PARTY HAS BEEN ADVISED OF THE POSSIBILITY OF SUCH DAMAGES.

END OF TERMS AND CONDITIONS